


colorcue

a publication for Compucolor users • v.3, # 2 • february 1980 • \$1



COLORCUE

**contributing
to the
success
of this issue**

USER INPUT

Cpt. DeFrance
Clarke

Judy Durant

Brian Hogan

Richard Manazir

Wayne Rebbechi

Dave Suits

TECHNICAL

ADVICE

Knox Pannill

Bruce Williams

ART DIRECTOR

Debbie Feldman

EDITOR

Cathy Abramson

menu

3 EDITOR'S LETTER

REM

4 DATA DIRECTORY

6 HIGH RESOLUTION GRAPHICS: 3-D BLOCKS

7 PASCAL'S TRIANGLE

9 TURKEY AND HUNTER ENHANCEMENT

SOFTWARE EXPLAINED

11 REENTERING PLOT SUBMODES

13 RANDOM VECTORS

PRODUCT SHOWCASE

14 PERSONAL INCOME TAX 1979 IS HERE!

14 FOUR EXCITING NEW GAMES TO
TICKLE YOUR FANTASY

KEEPING IT SIMPLE

16 COMPUCOLOR SCREEN COLOURS CODES

18 COLOR CHART

ADVANCED APPLICATIONS

19 DRAW

23 INPUT

25 ATTN/BREAK

27 SUBSCRIPTION AND BACK ISSUE ORDER FORM

ENCLOSURES

COMPUCOLOR PRICE LIST AND ORDER FORM

KEYBOARD LABELS ORDER FORM

CREARE SEMINAR REGISTRATION BROCHURE

— member international association of business communicators —

COLORCUE is published monthly for the users of the Compucolor II personal computer by Compucolor Corporation, P.O. Box 569, Norcross, Georgia 30071. Compucolor is a wholly owned subsidiary of Intelligent Systems Corp. Address all COLORCUE correspondence to our editorial office located at 100 Northcreek, Suite #250, 3715 Northside Parkway, N.W., Atlanta, Georgia 30327, (404) 261-3003. Subscriptions to COLORCUE are \$12.00 per year in the U.S., Canada, and Mexico, and \$24.00 elsewhere. © 1980 Compucolor Corporation. All rights reserved.

editor's letter

CONTEST! CONTEST! CONTEST!

**SEND US YOUR BEST DISPLAY AND WIN UP TO \$1,000
IN COMPUCOLOR MERCHANDISE!**

We've seen some spectacular graphics displayed on Compucolors at computer fairs recently -- not to mention the sophisticated and imaginative displays we have received in the mail -- and we're impressed! You are certainly Compucolor's best advertisers. Unfortunately, we only see the tip of the iceberg here at Compucolor and, naturally, we would like to see much, much more.

We are looking for displays that are visually exciting for use in Compucolor's nationwide ad campaign. Perhaps you have a striking chart or a practical graph, or maybe you've developed a 3-dimensional diagram or map. We are willing to offer a Grand Prize of \$1,000 in credit toward the purchase of Compucolor merchandise for the display that takes best advantage of Compucolor's superior color graphics! Use your credit to add to your Sof-Disk library, or expand your system with an extra disk drive, added memory, an upgraded keyboard -- whatever you choose!

And who knows? Maybe your display will appear in our next ad and earn you \$1,000 in Compucolor credit. Any display, other than the Grand Prize winner, used in a Compucolor ad, will entitle the developer to \$100 in Compucolor credit. At the same time, you will be promoting a machine that you believe in and your investment of programming time will result in more Compucolor owners and a stronger user group!

Send your displays to me, care of my editorial office. Keep a copy of your disk safe at home; Compucolor Corporation cannot be responsible for programs lost or damaged in the mail. **Deadline for all entries is April 30, 1980.** All disks become the property of Compucolor Corporation.

=

Now, here's the issue you've been waiting for -- the **GRAPHICS SPECIAL!!** I'd like to thank so many of you for writing in. Because color graphics is the feature that makes Compucolor so special, it's important to keep a fat file of potential **COLORCUE** articles handy. There will always be room

for your contributions! Please keep sending me your applications, graphics programs, and displays. Come on a pictorial tour of the plant with me -- next issue, the spotlight is on hardware.

=

rem

DATA DIRECTORY


by Judy Durant
Create, Inc.
Hanover,
New Hampshire

In spite of having invested several hundred thousand dollars in a DEC 11/70 and a DEC VAX 11/780, we still find considerable use for our Compucolor II. It is used for stand alone computational purposes as well as interactive data acquisition and control purposes in the test laboratory.

One of our applications is to read pressures, temperatures, rpm's and so on from installed instrumentation on a turbocharged automotive engine. Interfaced with the Compucolor is a data acquisition system, designed in-house.

Test data is stored on disk as a random file. Because one disk can store data for as many as 20 tests, keeping the data organized was one of the first problems we ran into. The solution was simple. We created our own directory file (not to be confused with the FCS Directory) to automatically track data for us. There is a random file on each data disk which consists of 20 records. Each file contains information for one test. When the main data acquisition program is run, it asks for a data file name, a test description, the date, and the operator's initials. This information is stored in the next available record of the file directory.

Additionally, the Compucolor keeps track of how much data was taken per engine set point, and how many set points were achieved while the experiment is running. This information is then written onto the directory file, to be assessed later when it's time to get the data off the disk for further reduction or plotting.





At first, the idea of two types of directories may seem redundant, but the one being described here is a more descriptive index than the FCS Directory. We have utilized it beyond its organizational function in our application. For example, when our hard copy plotting program is being run, the index is listed out and the user inputs the record number corresponding to the desired test. The program gets the random file name for that test and retrieves data from it using the recorded number of data points and set points. This automatically results in accurately dimensioned variable arrays. The description of the test, the date, and the operator are written onto the plot. All of this is accomplished by inputting one number.

Creating your own directory is not specific to a data acquisition application and can be useful to any CC user. There are lots of other possibilities for its use. If you have ideas or questions about our application, we'd enjoy hearing from you.

=

Creare Incorporated is a New Hampshire-based, high technology research and development engineering firm employing about ninety people. To contact Creare, call or write:

EDITOR'S NOTE

Creare, Inc.
Hanover, New Hampshire 03755
(603) 643-3800

Creare is presenting a two day seminar entitled "The Application of Low Cost Microcomputers for Data Acquisition and Control" in Detroit, Michigan, on March 25-26. See the enclosed brochure for further information.

=

**HIGH RESOLUTION
GRAPHICS: 3-D
BLOCKS**

Did you know that you can exceed the 128x128 graphics resolution of the Compucolor? You have to get a little fancy, and there are some limitations, but it's possible using the many special characters of the Compucolor.

As you know, the Compucolor character set contains 128 characters, all of which can be displayed in both single and double height. Each character is made up of a 6x8 dot matrix. If you can create a display with these special characters, an effective resolution of 384x256 can be realized (6x64=384, 32x8=256). The following program is a graphic example (pardon the pun) of what can be accomplished. The display it generates is from the Bay Area User Group in Mountain View, California, and made up almost entirely using special characters. You'll be surprised by the display this program creates. It stacks 3-D blocks of random heights on a 3-D grid. Here are some of the techniques used to produce this chart:

1. Line 140 creates a string consisting of the character containing a "right diagonal line" and the "cursor up" control character. The diagonal character can be created directly from the keyboard by typing (BGON)9,Shift (User), and has the ASCII value of 126. Notice that if D\$ is printed in succession, a continuous diagonal line is printed across the screen (PRINT D\$;D\$;D\$;D\$).

2. We create another string command in line 190 consisting of five characters of ASCII value 101 followed by an ASCII 122.

3. The real meat of this display is the subroutine starting at line 380. In this subroutine, lines 400 and 410 create the 3-dimensional blocks. Each value in these two PLOT statements is described below.

3,X,Y-K	places the cursor at location X,Y-K
29,16	sets foreground to black
30,C	sets background to color C
101	prints character
111	prints character
126	prints character
28	moves cursor up one line
26,26,26	moves cursor left 3 spaces
16	sets background to black (BG Flag is on)
29,C	sets foreground to color C
126	prints character
16	sets foreground to black (FG is on)
30,C	sets background to color C
32	prints character
126	prints character

Line 420 is used to "touchup" the bottom block in the random stack.

Each number, between the ones, is the sum of the two numbers on either side of it in the row above. Without going into the importance of the numbers found in this triangle, let it be pointed out that it has provided countless hours of pleasure to recreational mathematicians trying to find interesting relationships between the numbers in this triangle.

A pretty computer display can be made of the triangle if certain numbers are emphasized through special symbols. Suppose, for example, the triangle is printed using a '.' in the position of each even number and a '#' in the other positions.

```

          #
        # #
       # . #
      # # #
     # . . #
    # # . # #
   # . # . # #
  # # . # # #
 # # . # # #
# # . # # #

```

Other patterns are possible by considering those numbers that are multiples of 3, or multiples of 4, and so on. As more lines are added, the pattern becomes more interesting.

```

1 PLOT 12: REM CLEAR SCREEN
100 DIM A(60)
110 PRINT "ENTER SOME NATURAL NUMBER BIGGER THAN 1
      (0 TO QUIT)"
120 INPUT "N=";N : REM CHOOSE MULTIPLE TO BE EMPHASIZED
130 IF N=0 THEN PLOT 12:END
140 A$="." : REM PUT IN RED SPACES, RED HEARTS,
      BLUE CLUBS, ETC.
150 B$="#": REM PUT IN DIFFERENT CHARACTERS
160 PLOT 12: REM CLEAR SCREEN
170 PLOT 3,55,0: PRINT N: REM PUT N AT UPPER RIGHT
      OF SCREEN
180 A(1)=1
190 I=1
200 GOSUB 500: REM GRAPH LINE 1
210 A(2)=1
220 I=2
230 GOSUB 500: REM GRAPH LINE 2
240 FOR I=3 TO 30
250 A(I)=1
260 FOR J=I-1 TO 2 STEP -1
270 A(J)=A(J) + A(J-1)
280 IF A(J)>=N THEN A(J)=A(J)-N: REM PUT IN REMAINDER
290 NEXT
300 GOSUB 500: REM GRAPH LINE I
310 NEXT
320 PLOT 14:REM LARGE CHARACTERS
330 PLOT 3,0,0: REM MOVE CURSOR TO UPPER LEFT
340 GOTO 120: REM GET NEW VALUE OF N

```



```

500 REM SUBROUTINE TO GRAPH ROW OF TRIANGLE
510 PLOT 15: REM SMALL CHARACTERS
520 FOR K=1 TO I
530 PLOT 3,32-I+ 2*(K-1),I: REM PUTS EACH CHAR
    IN POSITION
540 IF A(K)=0 THEN PRINT A$:GOTO 560
550 PRINT B$
560 NEXT
570 RETURN

```

Note: Replacing lines 140, 150 with an INPUT statement would allow selection of characters interactively.

=

While Brian Hogan's program uses special characters, one of the most interesting things to do with the Triangle is to use Compucolor's colors to highlight the facets. **EDITOR'S NOTE**

An article printed in the Autumn 1979 issue of **COMPUTER PRODUCTS**, published by Creative Publications, used the Compucolor II to plot a beautiful modulo 8 triangle, with each number from 0 to 7 represented by a different color. Here's their version:

```

110 DIM A(31,31)
120 INPUT "WHAT MODULUS?";M
130 A(1,1)=1
140 A(2,1)=1
150 A(2,2)=1
160 FOR X=1 TO 31
170 PRINT:PRINT TAB (31-X);
180 FOR J=1 TO X
190 IF J=1 OR X=J THEN A(X,J)=1:GOTO 210
200 A(X,J)=A(X-1,J)+ A(X-1,J-1)
210 R=A(X,J)-INT(A(X,J)/M)*M
220 IF R=0 THEN GOTO 230
225 PRINT R;:GOTO 240
230 PRINT " ";
240 NEXT J
250 NEXT X

```

Thanks, Creative Publications. Just add color and try this yourself for a fascinating display!

=

I modified the Turkey and Hunter program on page 70 of the **Programming Manual**. Now it keeps score, leaves a red and blue track of Turkey and Hunter positions and a green fence around the playing area. **TURKEY AND HUNTER ENHANCEMENT**

By Cpt. DeFrance Clarke
of Fairborn, Ohio


```

0 REM TURKEY AND THE HUNTER
10 DEF FN R(X)= INT (X* RND (1))
40 PLOT 27,24,15,6,0,12
50 REM DRAW BORDER AROUND SCREEN
60 PLOT 6,2,30,8,14+96
71 FOR I= 2TO 63:PLOT 96+ 5:NEXT :REM TOP LINE
72 PLOT 27,10,96+ 15
73 FOR I= 2TO 31:PLOT 96+ 2:NEXT :REM RIGHT SIDE LINE
74 PLOT 27,24,96+ 13,3,62,31
76 FOR I= 62TO 1STEP - 1:PLOT 31+ 96,26,26:NEXT :REM BOTTOM
78 PLOT 96+ 12
80 FOR I= 30TO 1STEP - 1:PLOT 26,28,96+ 1:NEXT :REM LEFT SIDE
90 PLOT 3,64,0
95 BC= 1
100 REM SET UP GAME PARAMETERS
110 HX= 1:HY= 1:REM HUNTER INITIAL POSITION
120 TX= 32:TY= 16:REM TURKEY INITIAL POSITION
130 TS= 2:REM TURKEY SPEED
150 HC= 39:TC= 15
180 REM FN R RETURNS RANDOM INTEGER IN RANGE -X TO X
190 DEF FN R(X)= - X+ INT ((2* X+ 1)* RND (1))
200 REM MOVE CURSOR TO TURKEY'S OLD POSITION
201 IF VCTHEN PLOT 3,TX,TY,6,0:REM VISIBLE
202 IF BCTHEN PLOT 3,127,TX,TY,8
210 TX=TX+ FN R(TS):REM CHANGE TURKEY X POSITION
220 TY= TY+ FN R(TS)
230 IF TX< 1THEN TX= 0:ESC= 1
232 IF TX> 62THEN TX= 63:ESC= 1
234 IF TY< 1THEN TY= 0:ESC= 1
236 IF TY> 30THEN TY= 31:ESC= 1
240 PLOT 32:REM CLEAR TURKEY'S LAST POSITION
250 REM MOVE CURSOR TO TURKEY'S NEW POSITION
253 IF VCTHEN PLOT 3,TX,TY,6,TC
256 IF BCTHEN PLOT 3,127,TX,TY,TC
260 PLOT ASC ("T"):REM OUTPUT TURKEY SYMBOL
270 IF ESCTHEN ESC= 0:GOTO 1000
300 REM MOVE CURSOR TO HUNTER'S OLD POSITION
303 IF VCTHEN PLOT 3,HX,HY,6,0
306 IF BCTHEN PLOT 3,127,HX,HY,32
310 REM RANDOM SELECT HUNTER'S MOVE IN X OR Y DIRECTION
320 IF RND (1)> (ABS (TY-HY)+ 1)/ (ABS (TY- HY)+ ABS (TX- HX)+
2)GOTO 500
400 HY= HY+ SGN (TY- HY):REM MOVE TOWARD TURKEY IN
Y DIRECTION
410 GOTO 600
500 HX= HX+ SGN (TX- HX):REM MOVE TOWARD TURKEY IN
X DIRECTION
600 PLOT 32:REM CLEAR HUNTER'S LAST POSITION
700 REM MOVE CURSOR TO HUNTER'S NEW POSITION
703 IF VCTHEN PLOT 3,HX,HY,6,HC
706 IF BCTHEN PLOT 3,127,HX,HY,HC
710 PLOT ASC ("H"):REM OUTPUT HUNTER SYMBOL
720 IF HX= TXAND HY= TYTHEN 2000:REM HUNTER CATCHES TURKEY
800 GOTO 200
1000 REM TURKEY ESCAPES
1010 PLOT 27,27
1020 PLOT 6,1+ 64,8,10,10

```



```

1040 PRINT "TURKEY ESCAPES!!!!"
1045 TP= TP+ 1
1100 GOTO 2900
2000 REM HUNTER CATCHES TURKEY
2010 PLOT 27,27,6,64+ 4,8,10,10
2020 HP= HP+ 1
2040 PRINT "GOTCHA NOW, YOU TURKEY!!"
2900 PLOT 3,64,0
3000 FOR I= 1 TO 2000:NEXT :REM  DELAY FOR AWHILE
3005 PLOT 15
3010 PLOT 12,3,0,26
3020 PRINT "  HUNTER:  ";HP
3030 PRINT "  TURKEY:  ";TP
4000 GOTO 50

```

=

software explained

The following excerpt is from "An Introduction to Color Graphics on the Compucolor II -- Instructions and Ideas" by Dave Suits of Rochester, N.Y. (Copyright 1980 D. B. Suits) This publication is in the process of being published by Dave. He will offer it to us, for a price of course, when publication is complete.

EDITOR'S NOTE

If you exit a particular plot submode (point, vector, incremental point, etc.), you may directly enter any other plot submode. For example, you may draw a vector and then immediately enter the incremental point plot submode. The point which is incremented will be the X2,Y2 of the last vector.

REENTERING PLOT SUBMODES

By Dave Suits
Rochester Institute
of Technology
Rochester, N.Y.

```

5 REM  PROGRAM 6.22
6 REM  REENTRY TO PLOT SUBMODE
10 PLOT 6,6,12: REM  SET UP
20 PLOT 2: REM  GENERAL PLOT MODE
29 REM  DRAW A VECTOR FROM 0,0 TO 63,63
30 PLOT 0,0,242,63,63
40 PLOT 251: REM  INCREMENTAL POINT PLOT SUBMODE
50 FOR J=1 TO 10
60 PLOT 96: REM  INCREMENT WITHOUT PLOTTING
70 PLOT 6: REM  INCREMENT ONLY ONCE
80 NEXT
90 PLOT 255

```


You may even exit the plot mode entirely in order to change colors or to print text, then reenter any plot submode and continue where you left off. Add to the above program these lines:

```

100 PLOT 6,1: REM RED
110 PLOT 2: REM BACK TO GENERAL PLOT MODE
120 PLOT 251: REM BACK TO INCREMENTAL POINT PLOT SUBMODE
130 FOR J=1 TO 10
140 PLOT 96,6: REM CONTINUE THE DOTTED LINE
150 NEXT
160 PLOT 255
170 PLOT 6,2: REM GREEN
180 PLOT 2: REM REENTER GENERAL PLOT MODE
190 PLOT 242: REM VECTOR PLOT SUBMODE
200 PLOT 0,0: REM DRAW A VECTOR BACK TO THE STARTING
    POINT
210 PLOT 255

```

If the program does not specify the starting co-ordinate for a point or line, the starting point defaults to the last co-ordinate, which will be 0,0 if no other value has previously been used since the computer was turned on (or since you hit CPU RESET). Type in this:

```
PLOT 2,0,0,255
```

Now type in this program.

```

5 REM PROGRAM 6.23
6 REM PLOT MODE DEFAULT CO-ORDINATES
10 PLOT 6,6,12: REM SET UP
20 PLOT 2: REM GENERAL PLOT MODE
30 PLOT 251: REM INCREMENTAL POINT PLOT SUBMODE
40 FOR J=1 TO 20
50 PLOT 10: REM INCREMENT UP AND TO THE RIGHT
60 NEXT
70 PLOT 255

```

Now RUN it. RUN it again. And again. And again. Now hit CPU RESET, ESC E (BASIC reset) and RUN it again. Interesting. Or try adding these lines:

```

15 PLOT 2,0,0,255
17 FOR K=1 TO 6
.
.
.
80 PLOT 12
90 NEXT

```


The program below demonstrates the use of both the Random Number Operator and the vector plotting capability of your Compucolor II. **RANDOM VECTORS**

In this example, we use the RND function to obtain random X and Y coordinates on the screen. Computers always start counting at zero. Since the screen graphics resolution is 128x128, we need random values between 0 and 127. Therefore, the first position is the zero position and the last is the 127th position. The RND function always returns a random decimal value between 0 and 1. It will never return a value of exactly 0 or 1. We will always get a value from 0 to 127 by multiplying 128 by the RND function.

After calculating random coordinates and selecting a color, this program enters the plot mode (line 170) and then enters the vector plot mode (line 180). Once this has occurred, the program expects the next two values to be the X and Y coordinates respectively. The computer draws a line from the last X,Y position to the new X,Y position. The Compucolor assumes that everything sent to the monitor is an X,Y pair of coordinates until it sees a 255 (line 200) and exits the plot mode. We could provide X,Y pairs continuously and the Compucolor would continue to plot from the previous X,Y to the new X,Y.

Note that line 240 has the "RUN" command. When the program reaches this line, it merely starts over. Two versions of the program are included. Notice how much the program can be shortened by combining lines in the second version.

```
100 PLOT 6,2 : REM SET BLACK BACKGROUND AND
    GREEN FOREGROUND
110 PLOT 12 : REM CLEAR THE SCREEN
120 FOR I=1 TO 50 : REM LET'S DRAW 50 RANDOM LINES
130 X=RND(1)*128 : REM GET A RANDOM NUMBER BETWEEN
    0 AND 127
140 Y=RND(1)*128 : REM GET ANOTHER RANDOM NUMBER
    BETWEEN 0 AND 127
150 C=RND(1)*7+1 : REM GET A RANDOM COLOR-BUT NOT BLACK
160 PLOT 6,C : REM SET TO DRAW IN THE RANDOM COLOR
170 PLOT 2 : REM ENTER THE PLOT MODE
180 PLOT 242 : REM ENTER THE VECTOR GRAPHICS MODE
190 PLOT X,Y : REM DRAW A LINE TO NEW X,Y
200 PLOT 255 : REM EXIT THE PLOT MODE
210 NEXT I
220 REM
230 REM
240 RUN : REM WE'RE DONE, LET'S DO IT AGAIN
250 REM
300 REM ** A COMPRESSED VERSION OF THE ABOVE PROGRAM **
310 PLOT 6,2,12 : FOR I=1 TO 50 : X=RND(1)*128 : Y=RND(1)*128
320 C=RND(1)*7+1 : PLOT 6,C,2,242,X,Y,255 : NEXT I : RUN
READY
```


product showcase

PERSONAL INCOME TAX 1979 IS HERE!

Here's some timely help with your Income Tax return. Let the Compucolor do it for you. The program accepts inputs for FORM 1040, Schedule A and up to six W2 forms. If other forms or schedules must be completed, their results can be easily entered into the program. Its color-coded lines make entry quick and fool-proof. For ease of use, the program uses the same line numbers as the IRS forms. Up to five files of your tax data can be saved or retrieved.

Although the program does not contain the Tax Tables, it will calculate the amount upon which your tax will be based and direct you to the correct table. All you need to do is enter the tax from the appropriate tax table into the program.

Personal Income Tax must be run on a 16K or larger machine. The cost of the program is \$29.95 and it can be ordered through your nearest Compucolor dealer or directly from the factory using the enclosed parts order form.

=

FOUR EXCITING NEW GAMES TO TICKLE YOUR FANTASY

Compucolor introduces Lunar Lander, Air Raid, Sharks, and Shoot. Four new disks which offer something for everyone including some real challenges -- Lunar Lander, for example, and Sharks (guaranteed to drive you nuts!). Both of these games require dancing fingers to save you from disaster!

There's the old -- Towers, Kalah (which I used to know as the African Stone Game), and 15 Puzzle -- all based on ancient games. And there's the new -- Coalition and Hyper. Strategy games like Mill and Quintominoes will keep you going for hours. Each of these disks are being sold for \$19.95 and are available from your Dealer now.

AIR RAID (16K)

1. AIR RAID -- This is no drill, you've got radar confirmation: enemy bombers carrying nuclear weapons are approaching with fighter escorts. Man the anti-aircraft gun -- your country is depending on you to bring them down!

2. RACE -- Drive as fast as you can, but don't smash into the walls! Up to four players can compete, or race against the clock.

3. QUINTOMINOES -- Fitting the pieces of this puzzle

together is hard enough when you're playing by yourself, but when you've got an opponent to play against, good luck!

4. ROVER ROBOT -- Will Rover find a path to safety? He can provide limited information about the dangers that lie ahead, but it's up to you to decide his path -- and fate!

LUNAR LANDER (16K)

1. LUNAR LANDER -- It's been three days since you left Earth and it's time for your descent to the moon's surface. Unlike other lunar lander programs, this is a real time version. You don't have time to think -- you must act now before your fuel is depleted. There's just one more obstacle; you need to land with enough fuel to take off again or you will be stranded forever on the moon.

2. COALITION -- You are on your way to colonize a new planet and wish to set up a stable government as quickly as possible. Unfortunately, the planning council, of which you are a member, is split into two coalitions. Your decision could effect the winning coalition.

3. LINKO -- This intriguing game takes a little luck and a lot of skill. Try to form a link from one side of the screen to the other.

SHARKS (8K)

1. SHARKS -- Your midnight swim turns into a nightmare when you spot a shark's fin cutting through the waves. Frantically you swim for your life, but your splashing attracts even more sharks. The longer you evade the sharks in this interactive game, the higher you score. There is no winning: just see how long you can survive.

2. TOWERS -- Restacking the tower seems simple enough, but try it -- it's a real challenge. Or watch Compucolor II play and "learn from a master".

3. KALAH -- Planning ahead is what this ancient Indian game is all about. Originally played with stones and pits dug into the ground, you can play without getting your fingernails dirty! Play with a friend or match wits against the Compucolor II.

4. MILL -- A game of strategy for two players. If your motives are too obvious, you'll wind up losing. Outwit and capture your opponent by following one simple rule: play sneaky!

SHOOT (16K)

1. SHOOT -- Be the first to knock out your opponent's artillery. Both armies are down to their last cannon, so your

decisions on firing angle and blast force are crucial to winning.

2. 15 PUZZLE -- This is a modern version of a 100 year old game. The object is to arrange the 15 numbers in numerical order.

3. HYPER -- You'd better act fast in this real time space game because the enemy has already engaged you. Try different degrees of difficulty by changing the number and type of enemy.

4. SEAWAR -- Dive, Dive, Dive! Enemy ships have been spotted. Use your arsenal of torpedoes to destroy the enemy fleet. Be careful! Some of the ships are on your side.

=

keeping it simple

EDITOR'S NOTE

Thirteen year old Wayne Rebbechi from Wangaratta, Victoria, Australia has devised the following charts to help him remember the proper color codes to use. Wayne says that, as a beginning programmer, these charts have helped him alot. We think you'll agree. Thanks, Wayne!

COMPUCOLOR SCREEN STATIONARY FOREGROUND COLOURS CODES

By Wayne Rebbechi
of Wangaratta,
Victoria,
Australia

	BACKGROUND								
		B	R	G	Y	BL	M	C	W
F O R E G R O U N D	B	0	8	16	24	32	40	48	56
	R	1	9	17	25	33	41	49	57
	G	2	10	18	26	34	42	50	58
	Y	3	11	19	27	35	43	51	59
	BL	4	12	20	28	36	44	52	60
	M	5	13	21	29	37	45	53	61
	C	6	14	22	30	38	46	54	62
	W	7	15	23	31	39	47	55	63

BLINKING FOREGROUND

F O R E G R O U N D	BACKGROUND								
		B	R	G	Y	BL	M	C	W
	B	64	72	80	88	96	104	112	120
	R	65	73	81	89	97	105	113	121
	G	66	74	82	90	98	106	114	122
	Y	67	75	83	91	99	107	115	123
	BL	68	76	84	92	100	108	116	124
	M	69	77	85	93	101	109	117	125
	C	70	78	86	94	102	110	118	126
	W	71	79	87	95	103	111	119	127

PLOT 6,X WHERE X IS FOUND FROM THE TABLES ABOVE

COLOURS

B = BLACK

R = RED

G = GREEN

Y = YELLOW

BL = BLUE

M = MAGENTA

C = CYAN

W = WHITE

=

COLOR CHART

Did you ever wonder what colors go well together and which do not? In addition to demonstrating Compucolor's graphics, the following program creates a color chart with every possible color combination, including the 64 shades obtained by printing a null character on different color backgrounds. Two versions of the program are provided, a well-remarked version and a compressed version.

When the program is run, a number is shown above each of the 64 colored squares. In order to obtain the color shown, this number must follow a PLOT 6 command. For example:

PLOT 6,2 All printing will be green on a black background

Incidentally, the first version of the program takes up 1822 bytes of memory while the compressed version only uses 363 bytes. The decrease is largely, but not totally due to the elimination of the REMark statements. Every line in BASIC requires at least five bytes no matter what it contains. Those five bytes consist of a two byte pointer to the start of the next line, two bytes for the line number, and one byte, zero, indicating the line end. Does this mean that we can save five bytes every time we combine two lines? Well, almost. We take up one byte with every colon we use to separate our statements -- still a four byte savings. In addition to saving bytes, compressed programs run faster, so it pays to be economical.

```
100 REM ***** CREATE A COLOR CHART *****
110 PLOT 6,2: : REM SET FOREGROUND GREEN AND BACKGROUND
    BLACK
120 PLOT 12,15 : REM CLEAR THE SCREEN AND 1X CHARACTERS
130 PLOT 27,24 : REM SET SCREEN TO PAGE MODE (NO SCROLL)
140 REM
150 REM HERE ARE OUR EIGHT COLORS
160 DATA BLACK, RED, GREEN, YELLOW, BLUE, MAGENTA, CYAN, WHITE
170 REM
180 REM CREATE AN ARRAY WHICH CONTAINS THE NAMES OF THE EIGHT
    COLORS
190 FOR I=0 TO 7 : READ A$(I) : NEXT I
200 REM
210 REM WRITE THE EIGHT NAMES ACROSS THE TOP OF THE SCREEN
220 PLOT 27,10 : REM SET THE VERTICAL WRITE MODE
230 PLOT 6,6 : REM SET FOREGROUND CYAN AND BACKGROUND BLACK
240 FOR J=0 TO 7 : PLOT 3,J*7+11,0 : REM SET CURSOR TO TOP,
    START AT THE 11
250 REM THE POSITION, INCREMENT 7 EACH TIME
260 PRINT RIGHT$(" "+A$(J),7) : NEXT J : REM PRINT THE
    RIGHT 7 CHARACTERS
270 REM OF A STRING CONSISTING OF 5
280 REM BLANKS FOLLOWED BY THE NAME
290 REM
300 REM WRITE THE EIGHT NAMES FOLLOWED BY THE POSSIBLE
    COMBINATIONS
```



```

310 PLOT 27,24 : REM SET PAGE MODE
320 PLOT 6,3 : REM SET YELLOW FOREGROUND AND BLACK
    BACKGRD.
330 REM SET UP A LOOP FOR THE EIGHT COLORS
340 FOR I=0 TO 7 : PLOT 6,3
350 Y=I*3+8 : PLOT 3,0,Y+1 : PRINT A$(I) : REM PRINT THE COLOR
    NAME
360 REM PRINT EACH OF THE EIGHT COMBINATIONS FOR THE CURRENT
    COLOR
370 FOR J=0 TO 7
380 C=J+I*8 : X=J*7+9 : REM GET THE COLOR AND X COORD.
390 PLOT 3,X,Y : REM POSITION THE CURSOR
400 PRINT " ";C : REM PRINT COLOR NUMBER
410 PLOT 3,X,Y+1,6,C : REM POSITION CURSOR AND SET COLOR
420 PRINT " TEST " : REM PRINT THE WORD 'TEST' AT THE
    ABOVE LOC.
430 PLOT 3,X,Y+2 : REM POSITION THE CURSOR
440 PRINT "``````" : REM PRINT A NULL FGR ON THE SELECTED
    BGRD
450 NEXT J
460 NEXT I
470 GOTO 470 : REM WE'RE DONE LOOP FOREVER
480 REM

500 REM ** THE ABOVE PROGRAM IN A COMPRESSED FORM **
510 PLOT 6,2,12,15,27,24 : DATA BLACK,RED,GREEN,YELLOW,BLUE
    MAGENTA,CYAN,WHITE
520 FOR I=0 TO 7 : READ A$(I) : NEXT I : PLOT 27,10,6,6
530 FOR J=0 TO 7 : PLOT 3,J*7+11,0 : PRINT RIGHT$(" "+
    A$(J),7) : NEXT J
540 PLOT 27,24,6,3 : FOR I=0 TO 7 : PLOT 6,3
550 Y=I*3+8 : PLOT 3,0,Y+1 : PRINT A$(I)
560 FOR J=0 TO 7 : C=J+I*8 : X=J*7+9 : PLOT 3,X,Y : PRINT
    " ";C
570 PLOT 3,X,Y+1,6,C : PRINT " TEST " : PLOT 3,X,Y+2 : PRINT
    "``````" : NEXT J,I
580 GOTO 580

```

=

advanced applications

This month's assembly language advanced application has been submitted by Richard Manazir of Southwick, Massachusetts. The program allows you to select colors and draw pictures on the CRT by moving the cursor keys and typing "D" whenever you want a character to appear on the screen. It has been written for the V6.78 system software and the equates for the IO and OSTR routines should be changed to read 17C8 (for IO) and 182A (for OSTR) to accommodate V8.79, or you can use "UPDAT" from the Assembler disk to do it for you. See December/January's issue, "Assembly Language 3: Making Programs Compatible with V6.78 and V8.79 Versions", for further details on making both versions of the system software compatible. This program can be easily modified so let your imagination go and experiment.

EDITOR'S NOTE

By Richard Manazir
of Southwick, Mass.

```

DRAW ;
LF EQU 10 ;LINEFEED
FF EQU 12 ;FORM FEED
RIGHT EQU 25 ;CURSOR RIGHT
LEFT EQU 26 ;CURSOR LEFT
UP EQU 28 ;CURSOR UP
LO EQU 3392H ;OUTPUTS SINGLE CHARACTER
OSTR EQU 33F4H ;OUTPUTS A STRING OF CHARACTERS
INPCRT EQU 81C5H ;JUMP VECTOR #31
KBDFL EQU 81DFH ;HOLDS NUMBER OF JUMP VECTOR
FOR
; THE KEYBOARD
KBRDY EQU 81FFH ;KEYBOARD READY FLAG
CHAR EQU 96 ;DEFINES CHARACTER
;
;
; CIINIT - THE CHARACTER INPUT INITIALIZATION ROUTINE
; SETS UP THE PARAMETERS NECESSARY FOR THE
; 'CHRINT' AND 'CI' ROUTINES.
;
CIINIT: MVI A,31 ;SETUP JUMP VECTOR #31
STA KBDFL ;STORE IN KEYBOARD FLAG
MVI A,0C3H ;LITERALLY 'JMP'
STA INPCRT
LXI H,CHRINT;GET ADDRESS OF 'CHRINT' ROUTINE
SHLD INPCRT+1;PLACE ADDRESS AFTER 'JMP'
XRA A ;CLEAR ACCUMULATOR
STA KBRDY ;CLEAR KEYBOARD READY FLAG
STA CHARIN ;CLEAR TEMPORARY CHARACTER STORAGE
;
; DRAW - THIS ROUTINE WILL DRAW PICTURES ON THE CCII
; USING THE ASCII CHARACTER 96 AND USER INPUT.
;
MVI A,12 ;CLEAR SCREEN
CALL LO
LXI H,MESSAG;LOAD INSTRUCTIONS FOR COLOR SELECTION
CALL OSTR ;PRINT INSTRUCTIONS
CALL CI ;INPUT FOR COLOR SELECTION
CPI 82 ;R FOR RED
JZ CLR1 ;SET COLOR
CPI 71 ;G FOR GREEN
JZ CLR2 ;SET COLOR
CPI 89 ;Y FOR YELLOW
JZ CLR3 ;SET COLOR
CPI 66 ;B FOR BLUE
JZ CLR4 ;SET COLOR
CPI 77 ;M FOR MAGENTA
JZ CLR5 ;SET COLOR
CPI 67 ;C FOR CYAN
JZ CLR6 ;SET COLOR
CPI 87 ;W FOR WHITE
JZ CLR7 ;SET COLOR
NEXT: MVI A,12 ;CLEAR SCREEN AFTER SETTING
COLOR
CALL LO
DRAW: CALL CI ;GET CHARACTER
CPI LEFT ;MOVE LEFT

```



```

JZ      DRAW1      ;DRAW?
CPI     RIGHT     ;MOVE RIGHT
JZ      DRAW2      ;DRAW?
CPI     UP        ;MOVE UP
JZ      DRAW3      ;DRAW?
CPI     LF        ;MOVE DOWN
JZ      DRAW4      ;DRAW?
JMP     DRAW       ;JUMP FOREVER
DRAW1:  MVI        A,LEFT  ;MOVE CURSOR LEFT
        CALL      LO
        CALL      CI
        CPI       68      ;WANT TO DRAW? USE LETTER D
        JNZ       DRAW    ;NO? CONTINUE
        MVI       A,CHAR  ;YES? PUT CHARACTER ON SCREEN
        CALL      LO
        JMP       DRAW    ;CONTINUE
DRAW2:  MVI        A,RIGHT ;MOVE CURSOR TO RIGHT
        CALL      LO
        CALL      CI
        CPI       68      ;WANT TO DRAW? USE LETTER D
        JNZ       DRAW    ;NO? CONTINUE
        MVI       A,CHAR  ;YES? PUT CHARACTER ON SCREEN
        CALL      LO
        JMP       DRAW    ;CONTINUE
DRAW3:  MVI        A,UP    ;MOVE CURSOR UP
        CALL      LO
        CALL      CI
        CPI       68      ;WANT TO DRAW? USE LETTER D
        JNZ       DRAW    ;NO? CONTINUE
        MVI       A,CHAR  ;YES? PUT CHARACTER ON SCREEN
        CALL      LO
        JMP       DRAW    ;CONTINUE
DRAW4:  MVI        A,LF    ;MOVE CURSOR DOWN
        CALL      LO
        CALL      CI
        CPI       68      ;WANT TO DRAW? USE LETTER D
        JNZ       DRAW    ;NO? CONTINUE
        MVI       A,CHAR  ;YES? PUT CHARACTER ON SCREEN
        CALL      LO
        JMP       DRAW    ;CONTINUE
CLR1:  MVI        A,17    ;SELECT RED
        CALL      LO
        JMP       NEXT
CLR2:  MVI        A,18    ;SELECT GREEN
        CALL      LO
        JMP       NEXT
CLR3:  MVI        A,19    ;SELECT YELLOW
        CALL      LO
        JMP       NEXT
CLR4:  MVI        A,20    ;SELECT BLUE
        CALL      LO
        JMP       NEXT
CLR5:  MVI        A,21    ;SELECT MAGENTA
        CALL      LO
        JMP       NEXT
CLR6:  MVI        A,22    ;SELECT CYAN
        CALL      LO

```



```

        JMP      NEXT
CLR7:   MVI      A,23      ;SELECT WHITE
        CALL    LO
        JMP     NEXT
;
;   CHRINT - THE CHARACTER INTERRUPT ROUTINE IS VECTORED
;           TO FROM THE KEYBOARD INPUT ROUTINE THROUGH
;           THE JUMP VECTOR 'INPCRT' (#31). THE CHARACTER
;           FROM THE KEYBOARD INPUT ROUTINE IS IN
;           REGISTER E.
;
CHRINT:  PUSH    H          ;SAVE REGISTERS THAT WILL BE
USED
        PUSH    PSW
        LXI     H,CHARIN;GET ADDRESS OF TEMP CHARACTER STORAGE
        XRA     A          ;CLEAR A0
        CMP     M          ;TEST FOR 'CHARIN' = 0
        JNZ     CFIN       ;IF NOT ZERO, THEN IGNORE INPUT
        MOV     A,E        ;GET CHARACTER FROM E
        ANI     127        ;STRIP UPPER BIT FOR ASCII
        MOV     M,A        ;STORE IN 'CHARIN'
CFIN:   POP     PSW        ;RESTORE USED REGISTERS
        POP     H
        EI              ;ENABLE INTERRUPTS
        RET           ;RETURN FROM INTERRUPT
;
;   CI - THE CHARACTER INPUT ROUTINE GETS A CHARACTER FROM
;        THE TEMPORARY STORAGE LOCATION 'CHARIN', CLEARS
;        THE KEYBOARD READY FLAG AND RETURNS WITH THE
;        CHARACTER IN A. IF THERE IS NO CHARACTER IN
;        'CHARIN', THEN 'CI' WILL HANG AND WAIT FOR
;        A CHARACTER.
;
CI:      EI              ;ENABLE INTERRUPTS
        LDA     CHARIN    ;GET CHARACTER
        MOV     E,A       ;STORE CHARACTER IN REGISTER E
        CPI     0         ;HAVE A CHARACTER?
        JZ      CI        ;IF NOT, HANG FOR CHARACTER
        PUSH    PSW       ;SAVE CHARACTER
        XRA     A         ;CLEAR ACCUMULATOR
        STA     KBRDY     ;CLEAR KEYBOARD READY FLAG
        STA     CHARIN    ;CLEAR TEMPORARY STORAGE FOR NEXT CHAR
        POP     PSW       ;RESTORE CHARACTER
        RET           ;ECHO CHARACTER FROM MAIN ROUTINE
CHARIN:  DS      1         ;TEMPORARY CHARACTER STORAGE
MESSAG:  DB      8,15,21,'WHICH COLOR? (E.G.R=RED,B=BLUE)
',239
        END      CIINIT

```

=

input

From Mike Ungerman of Sugarloaf Shores, Florida:

Many graphic programs designed for the Radio Shack TRS-80 computer can be converted by the Compucolor II user with an overall improvement due to color and increased resolution.

One command used by TRS-80 Level II BASIC and not available to Compucolor is the "POINT" FUNCTION. It checks the TRS-80 screen memory to see if a particular graphics block is "set" (on or off). This function can be simulated in the Compucolor II through the use of the PEEK command to look at screen memory and see what character is at any given location of the screen. This is useful when running an interactive graphics program to see if the screen is occupied at a specific "X" and "Y" coordinate.

To check the Compucolor II screen memory to see if a given X and Y location has any character other than a space (ASCII 32) use the following statement:

```
IF PEEK(28672 + (2*X) + (128*Y)) <> 32 THEN line no.
```

This is the same as the TRS-80 statement:

```
IF POINT(X,Y) THEN line no.
```

=

Here's an alternative if you want to check and see if the plot mode bit is set instead of whether a space character exists:

EDITOR'S NOTE

```
IF PEEK (28673 + 2 * X + 128 * Y) >127 THEN line no.
```

The check for 32, the space character, could be a plot point and not a space if the plot bit has been set. Keep in mind that even checking to see if the plot mode bit is on does not guarantee that a point will be illuminated because the character byte could be zero.

=

And from Captain DeFrance Clarke of Fairborn, Ohio

Have you ever wanted to act on single characters typed at the keyboard without a RETURN after each one? I use the following lines of code in several of my programs:

```
10 KB = 33278
100 PRINT "ENTER CHARACTER"
110 POKE KB,0
120 C=PEEK(KB) : IF C= 0 GO TO 120
130 C$ =CHR$(C)
```

These statements work by putting the value zero into the RAM location where the Compucolor stashes the last keyboard character typed. Then we just wait in a loop for the character to change. Note that C is a number between 1 and 255 which is the ASCII code of the character typed. So if we want to find out if the left arrow key was pushed 200 IF C=26 GO TO 1000. The only characters which you cannot use are LF (down arrow) and AUTO, because the Compucolor does something else with them.

=

Trevor and Denise Taylor of Blacksburg, Virginia have enhanced the Star Trek game. They are most anxious to meet other Compucolor owners. They write:

We have added to Star Trek and now enjoy it even more. Changes include:

1. The directions for moving and shooting are now 0 to 360 degrees instead of 1-8.

2. A COMPUTER has been added. When invoked (Command 4), it gives the angle (in degrees) and the distance (as a fraction) from the Enterprise to any other part of the segment it is in. This is particularly useful for shooting Klingons with torpedoes. If a star or Klingon is in the way, the message "obstacle in path" is printed.

3. Docking at a starbase brings ALL equipment and devices "up" again immediately.

4. Equipment goes "down" with a slightly higher probability -- more of a challenge!

5. The Enterprise is now dark blue instead of green and is much easier to see.

Anyone interested could send us a blank diskette and we will copy our program onto it. The address is:

Trevor and Denise Taylor
200A Foxridge Apts.
Blacksburg, Virginia 24060

=

Here's a man who loves a good game of chess! Ray Kersch is looking for a challenging, updated version of his favorite game. Do you have one to offer? It's your move! Contact:

Ray Kersch
2650 John Street
Easton, Pennsylvania 18042
(215)252-3319

=

attn/break

Many users have installed a printer handshake modification, a modification which can result in a significant increase in printing speed. Compucolor can not endorse the modification. It is not part of our original design and, as you will see, can result in problems. **PRINTER PROBLEMS**

Several Compucolor users have written complaining about a lower case "y" being typed by their printer when using the handshake modification. We managed to duplicate the problem and put our engineers to work. This is a software problem which occurs because the modification pulls down an unused line on the keyboard scanner port. The lower case "y" appears if the line is pulled down while the keyboard is being scanned.

The solution is to disable the interrupts during printing. This is easily accomplished by doing an OUT 8,4. This disables everything except the real time clock, and locks out the keyboard. The interrupts are enabled again by using OUT 8,255. To accomplish this from BASIC, type:

```
100 PLOT 27,13 : REM SEND ALL OUTPUT TO THE RS232 PORT
110 OUT 8,4      : REM DISABLE INTERRUPTS
120 PRINT : EVERYTHING GOES TO THE PRINTER
```

```
130 OUT 8,255   : REM ENABLES ALL INTERRUPTS
140 POKE 33265,0: ROM SEND ALL OUTPUT TO THE MONITOR
```


Handshake Modification Since all the keys are disabled, the only way to stop printing is by hitting the CPU RESET key.

Another solution is to hold any two alphanumeric keys down simultaneously while printing. This causes the scanner to disregard the keyboard since a valid input is not available.

We have reprinted the printer handshake modification for those of you who are interested.

1. Tie Pin 9 of the J2 edge connector to UD1 Pin 4
2. Tie UD1 Pin 6 to UC1 Pin 3
3. Tie UC1 Pin 4 to UC1 Pin 10
4. Add a 10K 1/4 W resistor between UD1 - 4 and H2V

=

HOW TO SUBSCRIBE AND ORDER BACK ISSUES

New owners are entitled to a six month's free subscription to **COLORCUE** as soon as their warranty card has been returned. Following the first six month's free subscription, the subscription rate is \$12 per year in the U.S., Canada, and Mexico and \$24 throughout the rest of the world.

Each quarter, back issues are combined into one issue covering that period. Quarterly issues may be ordered for \$5 each. Once the quarterly issue is ready, individual issues will not be sold separately. During the current quarter, individual issues are available for \$2 each.

Below is a list of all back issues of **COLORCUE**. To order, fill out the attached order blank, enclose your check or money order (no cash, please), and mail your order to:

Cathy Abramson
COLORCUE
100 Northcreek, Suite #250
3715 Northside Parkway, N.W.
Atlanta, Georgia 30327

Please send me the issues I have checked below:

- ___ Volume 1, Nos. 1, 2, 3 October-December 1978
- ___ Volume 2, Nos. 1, 2, 3 January-March 1979
- ___ Volume 2, Nos. 4, 5 April-May/June 1979
- ___ Volume 2, No. 6, 7, 8 August-November 1979
- ___ Volume 3, No. 1 December/January 1980

Please enter my subscription for ___ year(s) beginning _____.

I have enclosed my check or money order in the amount of \$ _____. Thank you.

NAME: _____

ADDRESS: _____

CITY: _____ STATE: _____ ZIP: _____

HOW TO SUBSCRIBE AND ORDER BACK ISSUES

New owners are entitled to a six month's free subscription to COMPUCOLOR as soon as their warranty card has been returned. Following the first six month's free subscription, the subscription rate is \$12 per year in the U.S., Canada, and Mexico and \$25 throughout the rest of the world.

Each quarter, back issues are combined into one issue containing that period. Quarterly issues may be ordered for \$5 each. Since the quarterly issue is ready, individual issues will not be sold separately. During the current quarter, individual issues are available for \$2 each.


Below is a list of all back issues of COMPUCOLOR. To order, fill out the attached order blank, enclose your check or money order (no cash, please), and mail your order for:

COMPUCOLOR
100 Northcross, Suite 2250
Norcross, Georgia 30092
Atlanta, Georgia 30327

Volume	Issue	Period
Volume 1	No. 1	October-December 1975
Volume 2	No. 1	January-March 1976
Volume 2	No. 2	April-May/June 1976
Volume 2	No. 3	August-November 1976
Volume 3	No. 1	December/January 1977

Please enter a subscription for _____ year(s) beginning _____

I have enclosed my check or money order for \$_____ to pay for your _____



Compucolor® Corporation

Intecolor Drive
225 Technology Park/Atlanta
Norcross, Georgia 30092
Telephone 404/449-5996